# Maintainer: Muflone http://www.muflone.com/contacts/english/
# Maintainer: Caleb Maclennan <caleb@alerque.com>
# Contributor: PhotonX <photon89 [at] gmail.com>
# Contributor: Felix Yan <felixonmars@archlinux.org>
# Contributor: Shanto <shanto@hotmail.com>
# Contributor: Athurg <athurg@gooth.cn>
# Contributor: TDY <tdy@gmx.com>

pkgname=shutter
pkgver=0.99.4
pkgrel=1
pkgdesc="The feature-rich screenshot tool"
arch=('any')
url="https://shutter-project.org/"
license=('GPL')
depends=('xdg-utils' 'imagemagick' 'procps' 'librsvg' 'libwnck3'
         'perl-carp-always'
         'perl-file-basedir'
         'perl-file-copy-recursive'
         'perl-file-which'
         'perl-glib-object-introspection'
         'perl-goocanvas2'
         'perl-goocanvas2-cairotypes'
         'perl-gtk3'
         'perl-gtk3-imageview'
         'perl-json'
         'perl-json-maybexs'
         'perl-locale-gettext'
         'perl-moo'
         'perl-net-dbus'
         'perl-number-bytes-human'
         'perl-proc-processtable'
         'perl-proc-simple'
         'perl-readonly'
         'perl-sort-naturally'
         'perl-www-mechanize'
         'perl-x11-protocol'
         'perl-xml-simple'
         'perl-xml-simple'
         'pango-perl'
         'xdg-desktop-portal')
optdepends=('perl-image-exiftool: read and write EXIF data'
            'nautilus-sendto: send screenshots via mail'
            'perl-path-class: Imgur upload support'
            'perl-lwp-protocol-https: Imgur upload support'
            'perl-net-oauth: Imgur upload support'
            'bc: 3D Rotate and 3D Reflection plugins support'
            'libappindicator-gtk3: display the tray icon in some environments (like GNOME)')
source=("${pkgname}-${pkgver}.tar.gz"::"https://github.com/shutter-project/${pkgname}/archive/v${pkgver}.tar.gz")
sha512sums=('57f7fdc1066375942bd1a9db381a8ba6f24fafbd5e0853a4dd19016438ec2f3cb039ebe90038d282bc07f6e45af9cdbb6f180f55e43357231b41818b8555347b')

package() {
  cd "${pkgname}-${pkgver}"
  make install prefix="${pkgdir}/usr"
}
